
16 Linear Differential Equations with Constant Coefficients

This chapter provides examples showing how Maple V can be used to solve higher order linear differential equa-
tions with constant coefficient. Some of the examples illustrate how Maple V can be used to solve problems the
same way that one does when solving differential equations by hand. Other examples are solved using the Maple
V command dsolve. Generally, when you are solving these kind of problems you will probably use dsolve. We
have included the other procedures here to reenforce the methods that you learn in the traditional course and to
illustrate how Maple V can simplify complex problems that involve lengthly computations.

16.1 Solving Homogeneous Equations

The first example will be to solve a fourth order linear differential equation with constant coefficients. The steps
used here are the same as one might use when working a problem by hand. In this example we solve the following
differential equation.

d4

dx4
y( x )+ 3

d3

dx3
y( x )− 16

d2

dx2
y( x )+ 12

d
dx

y( x ) = 0

Subsequently, we wish to find the solution of the differential equation which satisfies the initial values:

y(0) = 2, y′(0) = 0, y′′(0) = 4, y′′′(0) = 0.

We enter the differential equation and assign it to the variable “deq1.”

> deq1 := diff(y(x),x$4)+3*diff(y(x),x$3)-16*diff(y(x),x$2)+

12*diff(y(x),x)=0;

deq1 :=
(
∂4

∂x4
y( x )

)
+ 3

(
∂3

∂x3
y( x )

)
− 16

(
∂2

∂x2
y( x )

)
+ 12

(
∂

∂x
y( x )

)
= 0

The characteristic equation is obtained by replacing the n th order derivative by rn.

> ceq1 := subs({diff(y(x),x$4)=r^4,diff(y(x),x$3)=r^3,diff(y(x),x$2)=r^2,
diff(y(x),x)=r},deq1);

ceq1 := r4 + 3 r3 − 16 r2 + 12 r = 0

The first step in solving the problem “deq1” is to find the roots of the characteristic equation.
Note: The answer or result provided by Maple V in the next two steps is a sequence of Maple objects separated
by commas. Such a construction is called an expression sequence.

> solve(ceq1,r);

0,1,2,−6

> croots1 := ";

croots1 := 0,1,2,−6

Once we know the characteristic roots we can obtain a basis for the general solution. When we enclose a se-
quence like 0, 1, 2, -6 with square brackets, [..], we create a Maple V object known as a list. A Maple V list
preserves order of data in the way you specify it to be. Next we apply a Maple function known as map to the
list [0, 1, 2, -6]. This Maple function applies the same operation to each component in a data structure. The map
command is a kind of ”advanced” Maple command and you may need to invoke

> ? map

in order to obtain help. We now apply map twice to obtain a base of solutions for the homogeneous solution.

> map( (x,y) -> x*y,[croots1],x);

[ 0, x,2 x,−6 x ]

> base1 := map(exp,");

base1 := [ 1, ex, e(2 x ), e(−6 x ) ]

We can now write the general solution:

> ygeneral1 := sum(c[i]*base1[i],i=1..4);

ygeneral1 := c1 + c2 ex + c3 e(2 x ) + c4 e(−6 x )

In order to solve the initial value problem we need to solve for the constants by solving the equations:

y(0) = 2, y′(0) = 0, y′′(0) = 4, y′′′(0) = 0.

Maple V commands that do this are as follows:

> y0 := subs(x=0,ygeneral1);

y0 := c1 + c2 e0 + c3 e0 + c4 e0

> dy0 := subs(x=0,diff(ygeneral1,x));

dy0 := c2 e0 + 2 c3 e0 − 6 c4 e0

> ddy0 := subs(x=0,diff(ygeneral1,x$2));

ddy0 := c2 e0 + 4 c3 e0 + 36 c4 e0

> dddy0 := subs(x=0,diff(ygeneral1,x$3));

dddy0 := c2 e0 + 8 c3 e0 − 216 c4 e0

> pcont1 := solve({y0=2,dy0=0,ddy0=4,dddy0=0},{c[1],c[2],c[3],c[4]});

pcont1 :=
{

c3 = 5
4
, c2 = −16

7
, c1 = 3, c4 = 1

28

}
Using the above result we can now write the particular solution to the initial value problem.

> y1p := subs(pcont1,ygeneral1);

y1p := 3− 16
7

ex + 5
4

e(2 x ) + 1
28

e(−6 x )

We will give an example to show you how to include the graph of the function, yc, within the worksheet. First
execute the following command:

> plot(y1p,x=0..3);

Maple V should respond by opening a window with a graph of the desired function. It will look something like
the following:

400

x

You may leave the window there while you execute other commands. Use the mouse to move the window to
another section of the screen. Note that you may change the size of the window by placing the cursor on the lower
right corner, holding down the left button on the mouse and moving the mouse. Try it. You can also copy the plot
to your Maple Worksheet. That’s how we got the graph above. All you do (on a UNIX machine with X-windows)
is:

1. Select edit in the menu bar of the plotting window with the left button on the mouse.

2. With the edit item open select the copy option, again with the left mouse button. This temporarily puts a
copy of the graph in the “clipboard.”

3. Move the cursor to some point in the Maple Worksheet, click the left

mouse button.

4. Select the edit item from the menu bar of the Maple Worksheet and select paste when the menu appears.

Try this now by attempting to put a copy of the plot you have in the window. If you are using a Windows base PC
or a Macintosh then simply do the usual copy and pasting between windows.

Hopefully you were successful. You can use this technique to include graphs in assignments. So practice this until
you can perform the procedure successfully. Get help if you’re having trouble.

The next example involves a characteristic equation with complex roots. Consider the following equation.

d4

dx4
y( x )− 5

d3

dx3
y( x )+ 3

d2

dx2
y( x )+ 19

d
dx

y( x )− 30 y( x ) = 0

Again we define the differental equation as the variable “deq2.”

> deq2 := diff(y(x),x$4)-5*diff(y(x),x$3)+3*diff(y(x),x$2)+19*diff(y(x),x)

-30*y(x) = 0;

deq2 :=
(
∂4

∂x4
y( x )

)
− 5

(
∂3

∂x3
y( x )

)
+ 3

(
∂2

∂x2
y( x )

)
+ 19

(
∂

∂x
y( x )

)
− 30 y( x ) = 0

Next, we create the characteristic equation.

> subs({diff(y(x),x$4)=r^4,diff(y(x),x$3)=r^3,diff(y(x),x$2)=r^2,

diff(y(x),x) = r,y(x)=1},deq2);

r4 − 5 r3 + 3 r2 + 19 r− 30 = 0

> ceq2 := ";

ceq2 := r4 − 5 r3 + 3 r2 + 19 r− 30 = 0

Next we find the characteristic roots.

> roots2 := solve(ceq2,r);

roots2 := 3,−2,2+ I,2− I

Since two of the roots are complex, the base will involve sine and cosine functions.

> base2 := [exp(3*x),exp(-2*x),exp(2*x)*cos(x),exp(2*x)*sin(x)];

base2 := [ e(3 x ), e(−2 x ), e(2 x ) cos( x ), e(2 x ) sin( x ) ]

The general solution is:

> ygeneral2 := sum(c[i]*base2[i],i=1..4);

ygeneral2 := c1 e(3 x ) + c2 e(−2 x ) + c3 e(2 x ) cos( x )+ c4 e(2 x ) sin( x )

Now, if we desire a solution of deq that satisfies the initial conditions,

y(0) = −2, y′(0) = 0, y′′(0) = 1, y′′′(0) = 14,

we proceed as follows.

> y0 := subs(x=0,ygeneral2);

y0 := c1 e0 + c2 e0 + c3 e0 cos(0 )+ c4 e0 sin(0 )

> y0 := simplify(");

y0 := c1 + c2 + c3



                
16 LINEAR DIFFERENTIAL EQUATIONS WITH CONSTANT COEFFICIENTS 27

> dy0 := simplify(subs(x=0,diff(ygeneral2,x)));

dy0 := 3 c1 − 2 c2 + 2 c3 + c4

> ddy0 := simplify(subs(x=0,diff(ygeneral2,x$2)));

ddy0 := 9 c1 + 4 c2 + 3 c3 + 4 c4

> dddy0 := simplify(subs(x=0,diff(ygeneral2,x$3)));

dddy0 := 27 c1 − 8 c2 + 2 c3 + 11 c4

> solve({y0=-2,dy0=0,ddy0=1,dddy0=14},{c[1],c[2],c[3],c[4]});

{
c2 = −37

85
, c3 = −13

17
, c4 = 52

17
, c1 = −4

5

}
> subs(",ygeneral2);

− 4
5

e(3 x ) − 37
85

e(−2 x ) − 13
17

e(2 x ) cos( x )+ 52
17

e(2 x ) sin( x )

We now use the Maple command dsolve to solve these same problems in one step.
The general solution for the first differential equation, “deq1”, can be obtained as follows:

> dsolve(deq1,y(x));

y( x ) = C1 + C2 ex + C3 e(2 x ) + C4 e(−6 x )

It is just as easy to find the solution of the initial value problem:

> dsolve({deq1,y(0)=2,D(y)(0)=0,D(D(y))(0)=4,D(D(D(y)))(0)=0},y(x));

y( x ) = 3− 16
7

ex + 5
4

e(2 x ) + 1
28

e(−6 x )

We will now solve the second problem rewriting it so as to illustrate the use of the D notation.

> deq2 :=(D@@4)(y)(x) - 5*(D@@3)(y)(x) + 3*(D@@2)(y)(x) + 19*D(y)(x)

-30*y(x) = 0;

deq2 := D(4 )( y )( x )− 5 D(3 )( y )( x )+ 3 D(2 )( y )( x )+ 19 D( y )( x )− 30 y( x ) = 0

> dsolve({deq2,y(0)= -2,(D(y))(0)=0,(D@@2)(y)(0)=1,(D@@3)(y)(0)=14},y(x));

y( x ) = − 4
5

e(3 x ) − 37
85

e(−2 x ) − 13
17

e(2 x ) cos( x )+ 52
17

e(2 x ) sin( x )

16.2 Solving Non Homogeneous Linear Equations

Next we illustrate how Maple can be used to find a particular solution of a non homogeneous linear equation by
the method of undetermined coefficients. We will find the general solution of the following equation:

d3

dx3
y( x )− 4

d2

dx2
y( x )+ d

dx
y( x )+ 6 y( x ) = ( x3 − 4 x+ 2 ) ex

First we enter the equation and the we proceed as if we were solving the equation by hand.

> deq3 := diff(y(x),x$3)-4*diff(y(x),x$2)+diff(y(x),x)+6*y(x)=(x^3 -4*x + 2)*exp(x);

deq3 :=
(
∂3

∂x3
y( x )

)
− 4

(
∂2

∂x2
y( x )

)
+
(
∂

∂x
y( x )

)
+ 6 y( x ) = ( x3 − 4 x+ 2 ) ex

The first step is to find the general solution of the associated homogeneous equation. Thus we obtain the char-
acteristic equation.

> ceqn3 :=subs({diff(y(x),x$3)=r^3,diff(y(x),x$2)=r^2,diff(y(x),x)=r,
y(x)=1},lhs(deq3))=0;

ceqn3 := r3 − 4 r2 + r+ 6 = 0

> croots3 := [solve(ceqn3)];

croots3 := [ 2,3,−1 ]

> base3 := map((s,t)->exp(s*t),croots3,x);

base3 := [ e(2 x ), e(3 x ), e(−x ) ]

Now the general solution of the homogeneous equation is

> yc3 := sum(c[i]*base3[i],i=1..3);

yc3 := c1 e(2 x ) + c2 e(3 x ) + c3 e(−x )

Next we find a particular solution by undetermined coefficients. In this example we define yp as a function of
x,a,b,c, and d.

> yp3 := (x,a,b,c,d) -> (a*x^3+b*x^2+c*x+d)*exp(x);

yp3 := ( x, a, b, c, d )→ ( a x3 + b x2 + c x+ d ) ex

This is substituted into the differential equation, deq3, and simplified and conditions on the constants a,b,c, and
d are obtained.

> deq3a := subs(y(x)=yp3(x,a,b,c,d),deq3);

deq3a :=
(
∂3

∂x3
%1

)
− 4

(
∂2

∂x2
%1

)
+
(
∂

∂x
%1

)
+ 6 %1 = ( x3 − 4 x+ 2 ) ex

%1 := ( a x3 + b x2 + c x+ d ) ex

> eq := simplify(expand(deq3a/exp(x)));

eq := 6 a− 6 a x− 2 b− 12 a x2 − 8 b x− 4 c+ 4 a x3 + 4 b x2 + 4 c x+ 4 d =
x3 − 4 x+ 2

> left := lhs(eq)-rhs(eq);

left := 6 a− 6 a x− 2 b− 12 a x2 − 8 b x− 4 c+ 4 a x3 + 4 b x2 + 4 c x+ 4 d− x3 + 4 x

− 2

We must collect all coefficients of the same power of x and equate to zero.

> left := collect(",x);

left := (4 a− 1 ) x3 + (−12 a+ 4 b ) x2 + (4− 8 b− 6 a+ 4 c ) x+ 6 a− 4 c− 2 b− 2

+ 4 d

Finally the constants a,b,c, and d can now be solved.

> solve({coeff(left,x^3)=0,coeff(left,x^2)=0,coeff(left,x)=0,coeff(left,x,0)=0}
,{a,b,c,d});

{
a = 1

4
, b = 3

4
, c = 7

8
, d = 11

8

}
The particular solution is given by

> ypart3 := subs(",yp3(x,a,b,c,d));

ypart3 :=
(

1
4

x3 + 3
4

x2 + 7
8

x+ 11
8

)
ex

The above function is a particular solution obtained by undetermined coefficients. The following steps repre-
sent a check.

> subs(y(x)=ypart3,deq3);

(
∂3

∂x3
%1

)
− 4

(
∂2

∂x2
%1

)
+
(
∂

∂x
%1

)
+ 6 %1 = ( x3 − 4 x+ 2 ) ex

%1 :=
(

1
4

x3 + 3
4

x2 + 7
8

x+ 11
8

)
ex

> simplify(");

2 ex − 4 ex x+ ex x3 = ( x3 − 4 x+ 2 ) ex

> collect(",exp(x));

( x3 − 4 x+ 2 ) ex = ( x3 − 4 x+ 2 ) ex

Thus the solution, ypart3, is indeed a solution.
The next example is an illustration of variation of parameters. In your class you probably use this method only

for second order equations. Maple allows you to work harder problems.

> deq4 := diff(y(x),x$3)-4*diff(y(x),x$2)+diff(y(x),x)+6*y(x) = cos(x)^2;

deq4 :=
(
∂3

∂x3
y( x )

)
− 4

(
∂2

∂x2
y( x )

)
+
(
∂

∂x
y( x )

)
+ 6 y( x ) = cos( x )2

As was seen above, a general solution for the homogeneous system assosiated with th equation is

yh := c1 e2x + c2 e−x + c3 e3 x.

Thus we seek functions u1(x), u2(x), u3(x) such that

yp = u1(x) e2 x + u2(x) e−x + u3 e3 x

is a particular solution. The system to solve is

e2 xu′1 + e−xu′2 + e3 xu′3 = 0,

2 e2 xu′1 − e−xu′2 + 3 e3 xu′3 = 0,

4 e2 xu′1 + e−xu′2 + 9 e3 xu′3 = cos(x)2.

Let us derive this method using Maple for this situation. We have already found a base of solutions for the
homogeneous equations, we reintroduce them now in the variable base.

> base := [exp(2*x),exp(-x),exp(3*x)];

base := [ e(2 x ), e(−x ), e(3 x ) ]

Now we set up the equations in the same way that we would do it if we were solving the equation by hand.

> U := [u[1],u[2],u[3]];

U := [u1, u2, u3]

> i := ’i’:yp := sum(U[i](x)*base[i],i=1..3);

yp := u1( x ) e(2 x ) + u2( x ) e(−x ) + u3( x ) e(3 x )

The problem is to solve for the functions u1, u2, and u3 so that yp satisfies deq4. We first must solve three
equations with three unknowns, the derivatives of the functions u1, u2, u3.

> i := ’i’:eq1 := sum(diff(U[i](x),x)*base[i],i=1..3)=0;

eq1 :=
(
∂

∂x
u1( x )

)
e(2 x ) +

(
∂

∂x
u2( x )

)
e(−x ) +

(
∂

∂x
u3( x )

)
e(3 x ) = 0

> i := ’i’: eq2 := sum(diff(U[i](x),x)*simplify(diff(base,x)[i]),i=1..3)=0;

eq2 := 2

(
∂

∂x
u1( x )

)
e(2 x ) −

(
∂

∂x
u2( x )

)
e(−x ) + 3

(
∂

∂x
u3( x )

)
e(3 x ) = 0

> i:= ’i’: eq3:= sum(diff(U(x),x)[i]*simplify(diff(base,x$2)[i]),i=1..3)=cos(x)^2;

eq3 := 4

(
∂

∂x
u1( x )

)
e(2 x ) +

(
∂

∂x
u2( x )

)
e(−x ) + 9

(
∂

∂x
u3( x )

)
e(3 x ) = cos( x )2

> i := ’i’:dsol := solve({eq1,eq2,eq3},{seq(diff(U(x),x)[i],i=1..3)});

dsol :=
{
∂

∂x
u1( x ) = − 1

3
cos( x )2

e(2 x )
,
∂

∂x
u2( x ) = 1

12
cos( x )2

e(−x )
,
∂

∂x
u3( x ) = 1

4
cos( x )2

e(3 x )

}
Thus we have found the derivatives of the unknown functions and now we integrate to find
the functions.

> i := ’i’:sol := {seq(int(lhs(dsol[i]),x)=int(rhs(dsol[i]),x),i=1..3)};

sol :=
{

u2( x ) = 1
60
( cos( x )+ 2 sin( x ) ) ex cos( x )+ 1

30
ex,

u3( x ) = 1
52
(−3 cos( x )+ 2 sin( x ) ) e(−3 x ) cos( x )− 1

78
1

( ex )3
,

u1( x ) = − 1
24
(−2 cos( x )+ 2 sin( x ) ) e(−2 x ) cos( x )+ 1

24
1

( ex )2

}
Now we substute back into yp to obtain the solution.

> yp := simplify(subs(sol,yp));

yp := 11
260

cos( x )2 − 3
260

cos( x ) sin( x )+ 97
1560

Once a formula is obtained for a solution it is easy to find numerical values for it and and its derivatives for
arbitrary values of x.

> eval(subs(x=0,yp));

163
1560

> eval(subs(x=0,diff(yp,x)));

−3
260

> eval(subs(x=0,diff(yp,x$2)));

−11
130

The command dsolve will yield the same solution if we use the same initial conditions.

> dsolve({deq4,y(0)=163/1560,D(y)(0)=-3/260,(D@@2)(y)(0)=-11/130},y(x));

y( x ) = 11
260

cos( x )2 − 3
260

cos( x ) sin( x )+ 97
1560

We can also substitute yp into deq4.

> simplify(subs(y(x)=yp,deq4));

cos( x )2 = cos( x )2

Exercises 16.2 The purpose of this exercise is to obtain the general solution of the following differential equation

d5

dx5
y(x)− 8

d4

dx4
y(x)+ 25

d3

dx3
y(x)− 50

d2

dx2
y(x)+ 84

d
dx

y(x)− 72 y(x) = x2e5 x

For convenience let L(y)(x) be defined by

L(y)(x) = d5

dx5
y(x)− 8

d4

dx4
y(x)+ 25

d3

dx3
y(x)− 50

d2

dx2
y(x)+ 84

d
dx

y(x)− 72 y(x)

1. Write the Maple V command that assigns the equation

L(y)(x) = 0

to the variable “deq”.

2. Write the Maple V commands that will allow you to write the characteristic polynomial for

L(y)(x) = 0.

3. Use Maple V to find the characteristic roots for

L(y)(x) = 0.

4. Write the general solution yh for
L(y)(x) = 0.

5. Guess the form of a particular solution yp of

L(y)(x) = x2 exp(5x)

to be used the in the method of undetermined coefficients.

6. Use Maple V to evaluate and to simplify
L(yp)(x)

with these undetermined coefficients.

7. Use Maple V to find yp (i.e. determine the coefficients).

8. Write the general solution to
L(y)(x) = x2 exp(5x).

9. Use the Maple V command dsolve to find the general solution to

L(y)(x) = x2 exp(5x).

10. Use the Maple command dsolve to find the solution of the initial value problem

L(y)(x) = x2 exp(5x), y(0) = 0, y′(0) = 0, y′′(0) = 0, y′′′(0) = 0, y(iv)(0) = 0.

Then plot the graph of this solution, in the interval [0,3], and paste it into your Worksheet.


